Mean Value Theorem for Derivatives

Objective

This lab assignment explores the hypotheses of the Mean Value Theorem. As a result of completing this assignment you will have a better understanding of the meaning of the MVT. In particular, you will be able to determine when the MVT does — and does not — apply.

Background

The Mean Value Theorem for Derivatives states that

If f is both
(1) continuous on a closed interval $[a, b]$ and
(2) differentiable on the open intervall (a, b),
then there is at least one number c in (a, b) with the property that

$$\frac{f(b) - f(a)}{b - a} = f'(c).$$

Geometrically, the MVT says that there is a point $(c, f(c))$ where the slope of the tangent line is equal to the slope of the secant line on the interval $[a, b]$.

The `MeanValueTheorem` command in the `Student[Calculus1]` package provides access to the MVT with visual or symbolic results. A more convenient interface is available with the `MeanValueTheorem` maplet.

Recall, from the Graphical Interpretation of the Limit lab, that the `piecewise` command can be used to enter a function given by different formulae on different parts of its domain.

Discussion

Enter, and execute, the following Maple commands in a Maple worksheet.

Example 1: Function Not Continuous

```
> restart;                                      # clear Maple's memory
> with( Student[Calculus1] );                   # load package
> f := x->piecewise( x< 1, x+1,                 # define function
>                    x>=1, x-1 );
> a := 0;                                       # left endpoint
> b := 3;                                       # right endpoint
> msec := ( f(b)-f(a) ) / ( b-a );              # slope of secant line
> secline := f(a) + msec*(x-a);                 # secant line
> plot( [ f(x), secline ], x=a..b,              # function and secant line
>       discont=true );
> MeanValueTheorem( f(x), a..b );               # hypotheses not satisfied
```

Example 2: Function Continuous, but Not Differentiable

```
> g := x -> abs(x);                             # define function
> a := -2;                                      # left endpoint
> b := 5;                                       # right endpoint
> msec := ( g(b)-g(a) ) / ( b-a );              # slope of secant line
> secline := g(a) + msec*(x-a);                 # secant line
> plot( [ g(x), secline ], x=a..b );            # function and secant line
> MeanValueTheorem( g(x), a..b );               # hypotheses not satisfied
```

Example 3: Function Continuous and Differentiable

```
> F:=theta -> sin(theta)^2 + sin(theta) + theta + 1;
> a := 0;                                        # left endpoint
> b := 3*Pi/2;                                   # right endpoint
> msec := ( F(b)-F(a) )/( b-a );                 # slope of secant line
> secline := F(a) + msec*(theta-a);              # secant line
> plot( [ F(theta), secline ], theta=a..b );     # function and secant line
> MeanValueTheorem( F(theta), a..b );            # plot shows 2 c's
> dF := D(F);                                    # derivative function
> eq := dF(c) = msec;                            # MVT equation
> C := solve( eq, {c} );                         # exact solutions
> evalf( C );                                    # ignore solution ∉ (a,b)
> fsolve( eq, c=1..2 );                          # find solution in [1,2]
> fsolve( eq, c=3..4 );                          # find solution in [3,4]
> MeanValueTheorem( F(theta), a..b,              # exact solutions
>                   output=points );
> MeanValueTheorem( F(theta), a..b,              # approximate solutions
>                   output=points, numeric=true );
```

Notes

(1) Example 3 concludes with three different methods for finding all points c guaranteed by the MVT. When the `solve` command is used it is necessary to verify that all solutions are found and to discard any solution that is not in the interval $[a, b]$. To use `fsolve` to find different solutions, identify disjoint intervals where each c can be found. Lastly, the `MeanValueTheorem` command with `output=points` attempts to find exact values for c, i.e., using `solve`. To encourage Maple to return approximate solutions, add both `output=points` and `numeric=true`.

(2) Example 3 should be explored with the `MeanValueTheorem` maplet. Note that the maplet does check the hypotheses of the MVT. If they are not satisfied, this is communicated to the user in a dialog box. Unfortunately, this message is not as clear as it could be. Also, in some instances, the maplet may stop working after it is run with a function that does not satisfy the hypotheses of the MVT. Because of this it is recommended that you check that the function satisfies the hypotheses of the MVT *before* beginning to use the maplet.

Questions

(1) Let $f(x) = |x^2 - x - 2|$. Determine if the Mean Value Theorem applies to f on the interval $[a, b] = [0, 3]$. If the MVT does not apply, state the hypothesis that is not satisfied. If the MVT does apply, identify all numbers c in the interval where $\dfrac{f(b) - f(a)}{b - a} = f'(c)$. In either case, include a graph that supports your conclusion.

(2) Repeat Question (1) for $S(\theta) = \sin(\theta) + \cos(2\theta)$ on the interval $[a, b] = [-\pi, \frac{\pi}{2}]$.

(3) Repeat Question (1) for $g(t) = t + t^{-1}$ on the interval $[a, b] = [-1, 2]$.

(4) A car is stationary at a toll booth at 8:14A.M.. Twenty-five minutes later the car is 32 miles down the road from the toll booth.

(a) Explain, using the MVT, why the car must have exceeded 65 miles per hour at some time in the twenty-five minutes after leaving the toll booth.

(b)–(d) Determine, using the MVT, if the car exceeded 70, 75, and 80 miles per hour at some time in the twenty-five minutes after leaving the toll booth. (*Three answers required.)*